

INTÉGRATION

Bernard Dupont

Bernard.Dupont@univ-lille1.fr


En économie, les problèmes posés par l'intégration portent sur le calcul d'intégrales définies (section 1), la recherche de primitives (section 2) et la détermination d'intégrales généralisées ou impropres (section 3). Dans chaque cas, la commande essentielle est **int** qui est le raccourci du nom de commande plus long **integrate**. Lui est associée la commande inerte **Int**, raccourci de **Integrate**, qui permet d'une part de améliorer la présentation des résultats et d'autre part doit être toujours utilisée dans des cas particuliers.

Maple sait calculer un nombre impressionnant d'intégrales mais il faut bien comprendre qu'il est limité par les difficultés techniques inhérentes au sujet. Il arrive que des résultats soient douteux et il est bon de connaître les méthodes de vérification (section 4, paragraphe 1) ou de suivre pas à pas les étapes de la résolution (section 4, paragraphe 2). On doit parfois guider le logiciel, soit en forçant une résolution par parties (section 4, paragraphe 3) ou par un changement de variable (section 4, paragraphe 4), deux méthodes bien connues qu'on peut invoquer aussi dans le paquetage dédié **IntegrationTools** (section 4, paragraphe 5).

Signalons qu'il existe un "tuteur" spécialisé pour les utilisateurs occasionnels auquel on accède par le chemin **Outils**→**Tuteurs**→**Calcul-Une variable**→**Méthodes d'intégration**, ou encore par l'instruction suivante :

```
> Student[Calculus1][IntTutor]();
```

S'affiche la fenêtre interactive suivante, très complète, qu'il suffit de remplir correctement.



Bien qu'une aide (Help) soit disponible en anglais, elle ne saurait se substituer à un travail d'apprentissage des compétences et subtilités de Maple dans le domaine de l'intégration.

▼ Intégrale définie

Soit Xp une expression intégrable dépendant de la variable libre x . On veut calculer son intégrale sur un intervalle fermé inclus dans le domaine de définition, soit par exemple $\int_a^b Xp \, dx$. On utilise la commande `int(Xp,x=a..b)` où **a** est la borne inférieure et **b** la borne supérieure de l'intervalle.

```
> restart;
xp1:=sin(x)^2;#expression à intégrer
int(xp1,x=0..Pi/2);#calcul de l'intégrale
xp2:=(2*x^2+x+3)*exp(x);#nouvelle expression à intégrer
int(xp2,x=0..1);#calcul de l'intégrale
```

$$\begin{aligned} xp1 &:= \sin(x)^2 \\ &\frac{1}{4} \pi \\ xp2 &:= (2x^2 + x + 3) e^x \\ &-6 + 5e \end{aligned}$$

Pour enjoliver la présentation, on combine la commande `int` à la forme dite "inerte" `Int`, qui empêche l'évaluation de l'intégrale.

```
> Int(sin(x)^2,x=0..Pi/2)=int(sin(x)^2,x=0..Pi/2);
```

$$\int_0^{\frac{1}{2} \pi} \sin(x)^2 \, dx = \frac{1}{4} \pi$$

Le fait que `Int` empêche l'évaluation ne signifie pas que Maple ne peut pas retourner le résultat. Il est possible de forcer l'évaluation par l'instruction `value` quand on veut un résultat exact et par l'instruction `evalf` quand on veut un résultat numérique. Ainsi, on peut utiliser conjointement `Int` et `value` :

```
> restart;
def:=Int(1/((x^(2/3)*(1-x)^(1/3))),x=0..1);#forme inerte
def=value(def);#égalisation de la forme inerte à son résultat exact
def=evalf(def);#égalisation de la forme inerte à son résultat numérique
```

$$\begin{aligned} def &:= \int_0^1 \frac{1}{x^{2/3} (1-x)^{1/3}} \, dx \\ &\int_0^1 \frac{1}{x^{2/3} (1-x)^{1/3}} \, dx = \frac{2}{3} \pi \sqrt{3} \end{aligned}$$

$$\int_0^1 \frac{1}{x^{2/3} (1-x)^{1/3}} dx = 3.627598728$$

Dernier point : l'habitude pousse à donner le label **I** à une intégrale. En Maple, il ne faut surtout pas assigner la lettre **I** à une intégrale car ce symbole est réservé pour exprimer l'imaginaire pur $\sqrt{-1}$.

```
> I:=int(x+exp(x),x=0..2);#assignation fautive
```

```
It:=int(x+exp(x),x=0..2);#assignation licite
```

Error, illegal use of an object as a name

$$It := 1 + e^2$$

(1.1)

Primitive

Soit Xp une expression dépendant de la variable libre x . On veut en déterminer une primitive, encore appelée intégrale indéfinie (en anglais : Indefinite Integral), soit $\int Xp dx$. On utilise la commande

int(xp,x) pour calculer directement l'intégrale et la forme inerte **Int(xp,x)** pour empêcher l'évaluation.

```
> restart;
```

```
P:=sin(x)^2;#expression dont on cherche la primitive
```

```
Int(P,x)=int(P,x);#égalisation de la forme inerte à la primitive
```

```
Q:=(a*x^2+b*x+c)*exp(x);#expression dépendant de paramètres dont on cherche la primitive
```

```
Int(Q,x)=int(Q,x);#égalisation de la forme inerte à la primitive
```

$$P := \sin(x)^2$$

$$\int \sin(x)^2 dx = -\frac{1}{2} \sin(x) \cos(x) + \frac{1}{2} x$$

$$Q := (ax^2 + bx + c) e^x$$

$$\int (ax^2 + bx + c) e^x dx = (c - b + 2a + bx - 2ax + ax^2) e^x$$

Notons que **Maple ne fait pas apparaître de constante d'intégration dans le résultat.**

Si Maple n'est pas parvenu à intégrer la fonction proposée par l'utilisateur, il renvoie la forme inerte.

```
> restart;
```

```
int(log(x)/(1-exp(x)),x);
```

$$\int \frac{\ln(x)}{1 - e^x} dx$$

Il peut arriver que Maple calcule l'intégrale demandée mais que le résultat retourné soit incompréhensible. Dans ce cas, il faut demander une évaluation de l'intégrale mais ce procédé est réservé aux intégrales définies.

```
> restart;
```

```
It:=int(1/(1-log(x)),x);#le résultat est abscons
```

```
value(It);#évaluation sans effet..
```

```
evalf(It);#cette évaluation n'améliore pas vraiment le résultat
int(1/(1-log(x)),x=1..2);#passage à une intégrale définie
evalf(%);#l'évaluation est possible
```

```
It := e Ei(1, 1 - ln(x))
      e Ei(1, 1 - ln(x))
      2.718281828 Ei(1., 1. - 1. ln(x))
      -Ei(1, 1) e + Ei(1, -ln(2) + 1) e
      1.820209823
```

Intégrale généralisée

Une intégrale généralisée se présente quand au moins une borne d'intégration n'appartient pas au domaine de définition de la fonction ou est infinie. Une telle intégrale est soit convergente (CV) soit divergente (DV). Maple est capable de répondre à un certain nombre de problèmes, mais pas tous. Il arrive même qu'il renvoie en échos des résultats mystérieux.

```
> restart;
f:=1/sqrt((1-x)*(x+1));It1:=int(f,x=-1..1);#les deux bornes
n'appartiennent pas à l'ensemble de définition
```

$$f := \frac{1}{\sqrt{(1-x)(x+1)}}$$

$$It1 := \pi$$

```
> It2:=int(sin(x)/x,x=0..infinity);#une borne est infinie
```

$$It2 := \frac{1}{2} \pi$$

```
> g:=1/(x^(1/3)*(1-x)^(1/3));It3:=int(g,x=0..1);#les deux bornes
n'appartiennent pas à l'ensemble de définition
```

$$g := \frac{1}{x^{1/3} (1-x)^{1/3}}$$

$$It3 := B\left(\frac{2}{3}, \frac{2}{3}\right)$$

Le dernier résultat est peu lisible, sauf si on connaît la définition mathématique de la fonction B , pour fonction Béta qui est définie à partir de la fonction Γ (Gamma) par $B(x, y) = \frac{\Gamma(x) \Gamma(y)}{\Gamma(x+y)}$. Ceci dit, on peut demander une évaluation numérique par **evalf**. Dans ce cas, le résultat est renvoyé avec 10 chiffres significatifs par défaut. S'il est nécessaire d'affiner la réponse, on donne en second argument de **evalf** le nombre de chiffres désiré.

```
> evalf(%);evalf(%,50);
2.053390218
2.0533902179391771810332309365990389868670514262072
```

Dans le cas où l'intégrale généralisée est DV, Maple retourne une réponse du type ∞ ou encore, dans les problèmes délicats, la forme non évaluée. Dans ce dernier cas, une demande d'évaluation renvoie un message d'erreur ou ... fait planter le système (d'où l'intérêt de sauvegarder régulièrement sa feuille de travail).

```
> restart;
I1:=int(log(x),x=1..infinity);#cas d'une intégrale DV
I1:= ∞
```

Approfondissements

Même si Maple est performant dans le calcul d'intégrales et de primitives, il lui arrive de se tromper ou de retourner des réponses mystérieuses ou encore de "sécher" sur un problème. Les doutes sur la fiabilité des résultats peuvent en partie être levés en recourant à la technique la plus simple de vérification : dériver la primitive et comparer le résultat à la fonction qu'il s'agissait d'intégrer. Une autre possibilité est de suivre pas à pas les étapes des calculs effectués par le logiciel. Par ailleurs, quand il pressent qu'il faut guider la résolution, l'utilisateur peut effectuer les techniques bien connues de changement de variables et d'intégration par parties soit en recourant aux ressources du paquetage **student** soit celles du paquetage **IntegrationTools**.

Vérification d'un résultat

Une bonne façon de vérifier qu'une primitive est correcte est de la dériver et comparer le résultat avec la fonction intégrande de départ. Pour avoir un résultat probant, il convient de bien maîtriser les techniques de simplification.

```
> restart;P:=x/(x^3+1);I1:=Int(P,x)=value(Int(P,x));
```

$$P := \frac{x}{x^3 + 1}$$

$$I1 := \int \frac{x}{x^3 + 1} dx = -\frac{1}{3} \ln(x + 1) + \frac{1}{6} \ln(x^2 - x + 1) + \frac{1}{3} \sqrt{3} \arctan\left(\frac{1}{3} (2x - 1) \sqrt{3}\right)$$

```
> diff(rhs(I1),x);normal(%,'expanded');
```

$$-\frac{1}{3(x+1)} + \frac{1}{6} \frac{2x-1}{x^2-x+1} + \frac{2}{3 \left(1 + \frac{1}{3} (2x-1)^2\right)} \frac{x}{x^3+1}$$

Demande d'informations sur les étapes du calcul

L'instruction **infolevel[*integrate*]:=N**, où **N** est un entier naturel qu'on égalisera de préférence à 2, fixe le niveau d'informations demandées et permet de visualiser les étapes de la résolution.

```
> restart;
infolevel[integrate]:=2;
Integrate(x/(x^5+2*x+1),x):
%=value(%);
```

```
infolevel[int]:=2
```

```
`int/indef1:`, `first-stage indefinite integration`
```

```
`int/ratpoly:`, `rational function integration`
```

```
`int/ratpoly/horowitz:` , integrating
```

$$\frac{X}{X^5 + 2X + 1}$$

```
`int/ratpoly/horowitz:` , `Horowitz' method yields`
```

$$\int \frac{X}{X^5 + 2X + 1} dX$$

```
`int/risch/ratpoly:` , `Rothstein's method - factored  
resultant is`
```

$$\left[\left[z^5 - \frac{500}{11317} z^3 + \frac{4}{11317} z + \frac{1}{11317}, 1 \right] \right]$$

```
`int/risch/ratpoly:` , `result is`
```

$$\sum_{R=\text{RootOf}(11317 Z^5 - 500 Z^3 + 4 Z + 1)} -R \ln \left(-X + \frac{65376045600}{64828679} -R^4 - \frac{4270752875}{64828679} -R^3 \right. \\ \left. - \frac{625000000}{64828679} -R^2 + \frac{447235682}{64828679} -R - \frac{21514240}{64828679} \right) \\ \int \frac{x}{x^5 + 2x + 1} dx = \sum_{R=\text{RootOf}(11317 Z^5 - 500 Z^3 + 4 Z + 1)} -R \ln \left(x + \frac{65376045600}{64828679} -R^4 \right. \\ \left. - \frac{4270752875}{64828679} -R^3 - \frac{625000000}{64828679} -R^2 + \frac{447235682}{64828679} -R - \frac{21514240}{64828679} \right)$$

Le texte qui précède l'affichage du résultat indique que l'automate a d'abord compris qu'il devait chercher une primitive ([indefinite integration](#)) d'une fonction rationnelle ([rational function integration](#)), puis a déclenché des méthodes spécifiques de factorisation, hors de portée de l'économiste moyen. Il est vrai que le dénominateur est particulièrement retors... L'output n'est guère explicite, mais on se convaincra que la réponse est juste en dérivant la primitive :

```
> normal(diff(rhs(%),x),`expanded`);
```

$$\frac{x}{x^5 + 2x + 1}$$

Intégration par parties

Soit U une fonction de dérivée u et V une fonction de dérivée v . On cherche $\int U v dx$. La méthode d'intégration par parties repose sur l'égalité $\int U v dx = UV - \int V u dx$. Maple effectue le calcul à l'aide de la commande **intparts** qui se trouve dans le paquetage **student**. Elle a pour syntaxe **intparts(It,U)** où **It** est une intégrale, définie ou indéfinie, écrite sous sa forme **inerte** et ayant pour forme $\int U v dx$; **U** est la fonction-expression à différencier. L'output est une forme **inerte** évaluable avec **value**.

Exemple d'une intégrale définie : soit à calculer $\int_0^1 x \arctan\left(\sqrt{\frac{1-x}{1+x}}\right) dx$. Il est clair qu'on a

tout intérêt à poser $U = \arctan\left(\sqrt{\frac{1-x}{1+x}}\right)$ et $v = x$.

```
> restart;
with(student):#chargement du paquetage student
U:=arctan(sqrt((1-x)/(1+x)));#U sera l'expression à
différencier
I1:=Int(x*U,x=0..1);#écriture de la forme inerte
intparts(I1,U);#intégration par parties de I1
value(%)#évaluation exacte
```

$$U := \arctan\left(\sqrt{\frac{1-x}{1+x}}\right)$$

$$I1 := \int_0^1 x \arctan\left(\sqrt{\frac{1-x}{1+x}}\right) dx$$

$$- \left(\int_0^1 \frac{1}{4} \frac{\left(-\frac{1}{1+x} - \frac{1-x}{(1+x)^2}\right) x^2}{\sqrt{\frac{1-x}{1+x}} \left(1 + \frac{1-x}{1+x}\right)} dx \right)$$

$$\frac{1}{16} \pi$$

Exemples d'intégrales indéfinies :

```
> intparts(Int(x^k*ln(x), x), ln(x));
value(%)
```

$$\frac{\ln(x) x^{k+1}}{k+1} - \left(\int \frac{x^{k+1}}{x(k+1)} dx \right)$$

$$\frac{\ln(x) x^{k+1}}{k+1} - \frac{x^{k+1}}{(k+1)^2}$$

```
> intparts(Int(sin(x)*x+sin(x), x), sin(x));
value(%)
```

$$\sin(x) \left(x + \frac{1}{2} x^2\right) - \left(\int \cos(x) \left(x + \frac{1}{2} x^2\right) dx \right)$$

$$\sin(x) \left(x + \frac{1}{2} x^2\right) - \cos(x) - \sin(x) x - \frac{1}{2} \sin(x) x^2 + \sin(x) - \cos(x) x$$

▼ Intégration par changement de variable

Le paquetage **student** renferme aussi la commande **changevar** qui, comme son nom l'indique, demande à Maple de procéder à un changement de variable dans le calcul d'une

intégrale définie ou d'une primitive.

On a la syntaxe `changevar(chv, It, u)` où `chv` explicite un changement de variable sous la forme simple $\phi(x) = u$ ou plus élaborée $h(x) = \phi(u)$ et où `It` est une intégrale ayant x pour variable d'intégration, définie ou indéfinie, écrite avec la forme inerte. L'output est une forme inerte qui montre bien comment l'intervalle d'intégration est redéfini.

```
> restart;
with(student):
Q:=(cos(x)+1)^3*sin(x);#écriture de l'expression intégrande
It:=Int(Q, x=a..b);#écriture de l'intégrale sous forme inerte
changevar(cos(x)+1=u, It, u);#changement de variable
```

$$Q := (\cos(x) + 1)^3 \sin(x)$$
$$It := \int_a^b (\cos(x) + 1)^3 \sin(x) dx$$
$$\int_{\cos(a) + 1}^{\cos(b) + 1} (-u^3) du$$

Le paquetage `IntegrationTools`

Le paquetage `IntegrationTools` porte bien son nom puisqu'il s'agit d'un regroupement d'outils spécifiques pour résoudre les problèmes d'intégration.

```
> restart;
with(IntegrationTools);
[Change, Combine, Expand, Flip, GetIntegrand, GetOptions, GetParts, GetRange,
GetVariable, Parts, Split, StripOptions] (4.5.1)
```

Les intégrales écrites sous forme inerte y sont automatiquement classées dans le type informatique `Integral`, subdivisé en deux sous-types : `DefiniteIntegral` et `IndefiniteIntegral`.

```
> It1:=Int(cos(x)*exp(x),x);#forme inerte d'une primitive ou
intégrale indéfinie
It2:=Int(exp(-rho*t)*U(c(t)),t=0..infinity);#forme inerte
d'une intégrale généralisée banale en contrôle optimal (rho
est le taux de préférence pour le présent, U est la fonction
d'utilité et c(t) la consommation instantanée)
type(It1,Integral);type(It2,Integral);#tests sur le type de
It1 et It2
type(It1,DefiniteIntegral);type(It1,IndefiniteIntegral);
#sous-tests sur It1
type(It2,DefiniteIntegral);type(It2,IndefiniteIntegral);
#sous-tests sur It2
```

$$It1 := \int \cos(x) e^x dx$$

$$It2 := \int_0^{\infty} e^{-pt} U(c(t)) dt$$

true

true

false

true

true

false

(4.5.2)

On récupère l'intégrande, la variable d'intégration ou l'intervalle d'intégration avec **GetIntegrand**, **GetVariable** et **GetRange** (l'output est "blanc" dans le cas d'une primitive).

```
> GetIntegrand(It1);GetIntegrand(It2);
   GetVariable(It1);GetVariable(It2);
   GetRange(It1);GetRange(It2);
```

$\cos(x) e^x$

$e^{-pt} U(c(t))$

x

t

$0.. \infty$

(4.5.3)

La commande **Split** permet d'appliquer la relation de Chasles

$$\int_a^b f(x) dx = \int_a^c f(x) dx + \int_c^b f(x) dx.$$

```
> Split(It2,1);#intégrale scindée en deux morceaux
   Split(It2,[1,2,3]);#intégrale scindée en 4 morceaux
```

$$\int_0^1 e^{-pt} U(c(t)) dt + \int_1^{\infty} e^{-pt} U(c(t)) dt$$

$$\int_0^1 e^{-pt} U(c(t)) dt + \int_1^2 e^{-pt} U(c(t)) dt + \int_2^3 e^{-pt} U(c(t)) dt + \int_3^{\infty} e^{-pt} U(c(t)) dt \quad (4.5.4)$$

Flip applique la formule $\int_a^b f(x) dx = -\int_b^a f(x) dx$.

```
> Flip(It2);
```

$$-\left(\int_{\infty}^0 e^{-pt} U(c(t)) dt \right)$$

(4.5.5)

Expand et **Combine** servent respectivement à développer et réduire une intégrale dans laquelle l'intégrande est ou devient une combinaison linéaire de fonctions.

```
> Int(3*exp(x)+5*G(x),x)=Expand(Int(3*exp(x)+5*G(x),x));
   #développement d'une intégrale
   a*Int(x^5,x=1..infinity)+b*Int(log(x),x=1..infinity)=Combine
   (a*Int(x^5,x=1..infinity)+b*Int(log(x),x=1..infinity));
```

#réduction d'une intégrale

$$\int (3 e^x + 5 G(x)) dx = 3 \left(\int e^x dx \right) + 5 \left(\int G(x) dx \right)$$
$$a \left(\int_1^{\infty} x^5 dx \right) + b \left(\int_1^{\infty} \ln(x) dx \right) = \int_1^{\infty} (b \ln(x) + a x^5) dx \quad (4.5.6)$$

Mais le paquetage **IntegrationTools** est particulièrement intéressant pour effectuer des changements de variable et des intégrations par parties, au point que les commandes dédiées du paquetage **student** sont déclarées déclassées (deprecated) par l'aide en ligne.

L'intégration par parties est prise en charge par **Parts(It, U)** où **It** est une intégrale, définie ou indéfinie, écrite sous sa forme inerte et ayant pour forme $\int U v dx$; et **U** est la fonction-expression à différencier.

> **Parts(It2, U(c(t)))**;

$$\lim_{t \rightarrow \infty} \left(-\frac{e^{-\rho t} U(c(t))}{\rho} \right) + \frac{U(c(0))}{\rho} - \left(\int_0^{\infty} \left(-\frac{e^{-\rho t} D(U)(c(t)) \left(\frac{d}{dt} c(t) \right)}{\rho} \right) dt \right) \quad (4.5.7)$$

Le changement de variable se fait avec **Change(It, eq)** où **It** est une intégrale définie ou indéfinie et **eq** est une équation explicitant le changement de variable.

> **Change(It1, exp(x)=u)**;

Change(It1, cos(x)=u);

$$\int \cos(\ln(u)) du$$
$$\int \left(-\frac{u e^{\arccos(u)}}{\sqrt{1-u^2}} \right) du \quad (4.5.8)$$

Exercices

Les exercices proposés dans ce chapitre sont sans prétention et n'ont d'autre objectif que celui de vérifier que les techniques de base sont bien comprises, surtout dans le domaine du calcul d'intégrales définies et de primitives. La plupart se trouvent dans l'ouvrage de Jean-Pierre Lecoutre et Philippe Pilibossian "Analyse 1. Exercices corrigés avec rappels de cours" (Masson, collection Mathématiques pour les sciences économiques, 1991), qui est une mine de bons exercices pour économistes. On conseille fortement de les faire d'abord "à la main" avant d'utiliser Maple.

Exercice M1

Calculer les intégrales définies suivantes :

1. $\int_1^2 (x-1)(x-2) dx$

2. $\int_0^1 (3\sqrt{x} - 4x) dx$

3. $\int_0^8 (\sqrt{2x} + \sqrt[3]{x}) dx$

$$4. \int_{-2}^{-1} x^{-3} dx$$

$$5. \int_{-1}^{+1} \left(x^2 - \sqrt{|x|} \right) \sqrt[3]{x^2} dx$$

$$6. \int_a^b \left(\frac{1}{\sqrt{x}} - x^2 \right) dx$$

$$7. \int_2^0 \sqrt{|1-x|} dx$$

$$8. \int_0^1 \frac{1}{1+x} dx$$

$$9. \int_{-\infty}^{+\infty} e^t dt$$

$$10. \int_2^3 a^x dx \text{ pour } a \in \mathbb{R}_+^* - \{1\}$$

$$11. \int_{-1}^{+1} e^{-2x+1} dx$$

Exercice M2

L'objectif de cet exercice est de calculer l'intégrale définie $\int_{-2}^2 (x^3 - \sqrt[3]{x}) x^2 dx$.

1. Calculer directement l'intégrale avec l'instruction **int**.
2. Critiquez le résultat renvoyé par Maple.
3. Comment obtenir le résultat juste avec Maple?

Exercice M3

Calculer les intégrales définies suivantes à l'aide d'un changement de variable :

$$1. \int_1^5 \sqrt{2x-1} dx$$

$$2. \int_0^3 (25-3x)^{-\frac{1}{2}} dx$$

$$3. \int_0^2 \frac{x^2}{2+x^3} dx$$

$$4. \int_1^2 \frac{1}{x(1+\ln(x))} dx$$

$$5. \int_2^3 \frac{x + \frac{1}{2}}{x^2 + x - 3} dx$$

$$6. \int_0^1 \frac{e^x}{(10 - 3e^x)^2} dx$$

$$7. \int_0^4 \frac{1}{1 + \sqrt{x}} dx$$

$$8. \int_0^1 \frac{e^x - e^{-x}}{e^x + e^{-x}} dx$$

$$9. \int_0^1 x^2 \sqrt{1 + x^3} dx$$

$$10. \int_{\ln(3)}^{\ln(8)} \sqrt{e^x + 1} dx$$

Exercice M4

Calculer par parties les intégrales définies suivantes :

$$1. \int_2^3 \ln(x^2 - 1) dx$$

$$2. \int_0^1 \frac{x + 1}{e^x} dx$$

$$3. \int_0^1 x 2^x dx$$

$$4. \int_1^2 \ln\left(\frac{x}{x+1}\right) dx$$

$$5. \int_{-1}^{+1} (x + e^x)(2x - e^{-x}) dx$$

$$6. \int_2^3 (x+1) a^{x+1} dx \text{ avec } a \in \mathbb{R}_+^* - \{1\}$$

$$7. \int_2^1 x [\ln(x)]^2 dx$$

$$8. \int_1^4 \frac{\ln(x)}{\sqrt{x}} dx$$

Exercice M5

Donner les primitives des fonctions suivantes :

$$1. f(x) = \frac{1 + x^5 - x^6}{1 - x}$$

$$2. f(x) = \frac{3x^2 + 1}{x^2(x^2 + 1)^2}$$

$$3. f(x) = \frac{x^{3+1}}{x^2 - x + 2}$$

$$4. f(x) = \frac{1}{x^2 - 5x + 4}$$

$$5. f(x) = \frac{x}{x^3 - 3x + 2}$$

$$6. f(x) = \frac{1}{x^3 - 7x + 6}$$

$$7. f(x) = \frac{7x - 5}{x^3 + x^2 - 6x}$$

$$8. f(x) = \frac{ax + b}{(x - c)^2}$$

$$9. f(x) = \frac{x^2 + 6x + 5}{x^2 - 6x + 5}$$

$$10. f(x) = \frac{2x + 1}{(x^2 - 1)^2}$$

$$11. f(x) = \frac{4x^{2-6x+1}}{2x^3 - x^2}$$

$$12. f(x) = \frac{2x^4 + 1}{x^3(x^2 + x + 1)}$$

Exercice M6

Donner les primitives des fonctions suivantes :

$$1. f(x) = \frac{x - 1}{\sqrt{2x + 1}}$$

$$2. f(x) = \frac{x^2}{\sqrt{x - 1}} + 2x\sqrt{x - 1}$$

$$3. f(x) = \frac{x^2\sqrt{x}}{1 + \sqrt{x}}$$

$$4. f(x) = \frac{x}{2 + \sqrt{1 + x}}$$

$$5. f(x) = \frac{x}{\sqrt[3]{1 + x} - \sqrt{1 + x}}$$

$$6. f(x) = \frac{x + \sqrt{x+2} + \sqrt{x+1}}{\sqrt{x+2} + \sqrt{x+1}}$$

$$7. f(x) = \frac{1}{\sqrt{x^2 + x + 1}}$$

$$8. f(x) = \frac{1}{(1+x^2)^{\frac{3}{2}}}$$

$$9. f(x) = \frac{1}{x^2 \sqrt{x^2 - x + 1}}$$

$$10. f(x) = \frac{1}{(x-1)\sqrt{-x^2 + 3x - 2}}$$

$$11. f(x) = \frac{\sqrt{1+x^2} + 1}{\sqrt{1+x^2} - 1}$$

$$12. f(x) = \frac{x}{(1+x^2)\sqrt{1-x^4}}$$

Exercice M7

Donner les primitives des fonctions suivantes :

$$1. f(x) = \frac{\sqrt{x} + \ln(x)}{x}$$

$$2. f(x) = \frac{1}{\sqrt{(1+x^2) \ln(x + \sqrt{1+x^2})}}$$

$$3. f(x) = \frac{[\ln(x)]^2}{x^2}$$

$$4. f(x) = \frac{(a^x - b^x)^2}{a^x b^x}$$

$$5. f(x) = \frac{1}{2^x + 3}$$

$$6. f(x) = \frac{e^{2x}}{\sqrt{e^x + 1}}$$

$$7. f(x) = \frac{1}{e^{2x} + e^x - 2}$$

$$8. f(x) = \left(e^{\frac{x}{a}} + e^{-\frac{x}{a}} \right)^2$$

$$9. f(x) = e^x \sqrt{a - be^x}$$

$$10. f(x) = e^{\sqrt{x}}$$

$$11. f(x) = \sqrt{1 - e^{-2x}}$$

$$12. f(x) = \left[\ln(x + \sqrt{1+x^2}) \right]^2$$

$$\begin{cases} 13. f(x) = x^2 \ln(\sqrt{1-x}) \\ 14. f(x) = |x| \end{cases}$$

▼ Exercice M8

Soit l'intégrale définie :

$$I_n = \int_{-1}^{+1} (x^2 - 1)^n dx \text{ où } n \text{ est un entier naturel.}$$

1. Calculer I_0, I_1, I_2, I_3 .
2. Trouver une relation de récurrence entre I_n et I_{n-1} .
3. Mettre alors au point une procédure pour calculer I_n (n quelconque).